/*

Jacqueline Rifkin	
Proposal title: Introducing a Novel Framework for Understanding The Relationships Between Busyness, Idleness, and Happiness	

HYPOTHESES

Stated-Hyp1: "When people have a longer future time perspective, busyness will be experienced as
productive and idleness will be experienced as boring; whereas, when people have a
shorter future time perspective, idleness will be experienced as time for the self and
busyness will be experienced as stressful."

	Test-Hyp1: The experience of productivity will be higher in busy condition vs. idle condition; this effect will be moderated by time perspective (LONGER time perspective intensifies effect)

	Test-Hyp2: Tthe experience of boredom will be lower in busy condition vs. idle condition; ; this effect will be moderated by time perspective (LONGER time perspective intensifies effect)	
	
	
	Test-Hyp3: The experience of time for self will be higher in idle condition vs. busy condition; this effect will be moderated by time perspective (SHORTER time perspective intensifies effect)

	Test-Hyp4: Tthe experience of stress will be higher in busy condition vs. idle condition; ; this effect will be moderated by time perspective (SHORTER time perspective intensifies effect)		
	
	

Stated-Hyp1: "These shifts in subjective experience will impact well-being: When people have a
longer future time perspective, busyness will generate more happiness than idleness,
while the reverse will be true for shorter time horizon."	
	
	Test-Hyp5: There will be an interaction effect between time perspective and busy vs. idle condition on subjective wellbeing (longer time perspective *business will have positive effect)


	
*/

clear all
use Rifkin1073.dta, clear

********************************************************************************

*EXPERIMENTAL CONDITION

	* R was made to think about ways in which they were more/less busy
	tab BUSY
	recode BUSY (1=1 "1. busy") (2=0 "0. idle"), gen(busy_vs_idle)
	tab busy_vs_idle
	
	
* OUTCOMES
	* productivity
	tab Q3B
	clonevar productivity = Q3B 
	replace productivity=. if productivity>7
	
	* boredom
	tab Q4A
	clonevar boredom = Q4A
	replace boredom=. if boredom>7
	
	* stress
	tab Q2A 
	clonevar stress=Q2A
	replace stress=. if stress>7
	
	* time for self
	tab Q5A //[I feel like I have time to think about myself.] 
	tab Q5B // [I feel like I am able to consider my own needs.] 

	clonevar timeself1 = Q5A 
	clonevar timeself2 = Q5B
	replace timeself1=. if timeself1>7
	replace timeself2=. if timeself2>7	
	gen timeforself= timeself1 + timeself2
	drop timeself*
	tab timeforself
	
* subjective wellbeing
	tab Q6 // Overall, how happy are you feeling?
	tab Q7 // Overall, how fulfilled are you feeling?
	
	replace Q6=. if Q6>100
	replace Q7=. if Q7>100	
	
	gen swb = Q6+Q7
	sum swb
	
* MODERATORS
	
* measures of time perspective (2 items)
	/* recode missing.
	also recode so that higher values indicate longer future time perspectives
	*/

	* [My future is filled with possibilities.] How true is this of you?
		// higher values indicate longer time perspective
	tab Q8A
	gen time1 = Q8A
	replace time1 = . if Q8A>7
	tab time1
	
	* [I experience time as limited.] How true is this of you?
		// currently, higher values indicate shorter time perspective: reverse code
	tab Q8B
	replace Q8B=. if Q8B>7
	gen time2 = 8-Q8B
	tab time2
	
	gen timeperspective = time1 +time2
	tab timeperspective
	
	* gen a variable for short time perspective (makes testing easier)
	gen shorttimeperspective=15-timeperspective
	tab shorttimeperspective
	
********************************************************************************

* ANALYSIS

*Test-Hyp1: The experience of productivity will be higher in busy condition vs. idle condition; this effect will be moderated by time perspective (LONGER time perspective intensifies effect)

	reg productivity c.timeperspective##i.busy_vs_idle
	// reject. 0.685
	tess timeperspective#1.busy_vs_idle +, init(Rifkin1073) bonf(3)
	
*Test-Hyp2: Tthe experience of boredom will be lower in busy condition vs. idle condition; ; this effect will be moderated by time perspective (LONGER time perspective intensifies effect)	
	
	reg boredom c.timeperspective##i.busy_vs_idle
	// reject. 0.299	
	tess timeperspective#1.busy_vs_idle -, bonf(3)
	
*Test-Hyp3: The experience of time for self will be higher in idle condition vs. busy condition; this effect will be moderated by time perspective (SHORTER time perspective intensifies effect)

	reg timeforself c.shorttimeperspective##ib1.busy_vs_idle
	// reject. 0.068	
	tess shorttimeperspective#0.busy_vs_idle +, bonf(2)
	
*Test-Hyp4: Tthe experience of stress will be higher in busy condition vs. idle condition; ; this effect will be moderated by time perspective (SHORTER time perspective intensifies effect)		
	
	reg stress c.shorttimeperspective##i.busy_vs_idle
	// reject. 0.032 (bonferonni correction needed)		
	tess shorttimeperspective#1.busy_vs_idle +, bonf(2)
	
*Test-Hyp5: There will be an interaction effect between time perspective and busy vs. idle condition on subjective wellbeing (longer time perspective *business will have positive effect)

	reg swb c.timeperspective##i.busy_vs_idle
	// reject. 0.261
	tess timeperspective#1.busy_vs_idle +, bonf(3)

